
The Fast Fourier Transform

The Fourier transform was originally developed by Joseph Fourier [3] for the study of heat
transfer and vibrations. Fourier transforms are currently used in the study of differential
equations, approximation theory, quantummechanics, time-series analysis, implementation
of high precision arithmetic, digital signal processing, GPS, sound and video compression,
digital telephony and encryption. The fast Fourier transform is a divide and conquer
algorithm developed by Cooley and Tukey [1] to efficiently compute a discrete Fourier
transform on a digital computer. In 2000 Dongarra and Sullivan listed the fast Fourier
transform among the top 10 algorithms of the 20th century [2].

Joseph Fourier of École Polytechnique, James Cooley of IBM Watson Laborato-
ries and John Tukey of Princeton University and Bell Laboratories.

The Discrete Fourier Transform

The discrete Fourier transform is given by the matrix-vector multiplication Ax where A is
an N ×N matrix with general term given by akl = e−i2πkl/N with k = 0, 1, . . . , N − 1 and
l = 0, 1, . . . , N − 1. While standard mathematical notation for matrices and vectors use
index variables which range from 1 to N , we have shifted the indices by one so that the
first column and first row of A are given by k = 0 and l = 0. Shifting indices in this way
is both the natural for the C programming language and the mathematics. This shifted
notation for indices will be used throughout our computational study of linear algebra.

Define A to be the matrix whose entries are exactly the complex conjugates of the
entries of A. Our first result is

The Fourier Inversion Theorem. Let A be the N×N Fourier transform matrix defined
above. Then

A−1 =
1

N
A.

To see why this formula is true we first prove

The Orthogonality Lemma. Suppose l, p ∈ {0, 1, . . . , N − 1}, then
N−1∑
q=0

ei2π(l−p)q/N =
{
N for l = p
0 otherwise.

Proof of The Orthogonality Lemma. Since

0 ≤ l ≤ N − 1 and − (N − 1) ≤ −p ≤ 0,

then −(N − 1) ≤ l − p ≤ N − 1 and consequently

−2π
(
1− 1

N

)
≤ 2π(l − p)/N ≤ 2π

(
1− 1

N

)
.

Define ω = ei2π(l−p)/N . Since the only time eiθ = 1 is when θ is a multiple of 2π, we
conclude that

ω = 1 if and only if l = p.

Clearly, if l = p then

N−1∑
q=0

ei2π(l−p)q/N =
N−1∑
q=0

wq =
N−1∑
q=0

1 = N.

On the other hand, if l ̸= p then ω ̸= 1. In this case,

ωN = ei2π(l−p) = 1

and the geometric sum formula yields that

N−1∑
q=0

ei2π(l−p)q/N =

N−1∑
q=0

ωq =
1− ωN

1− ω
=

1− 1

1− ω
= 0.

This finishes the proof of the lemma. ////

We are now ready to explain the Fourier inversion theorem.

Proof of The Fourier Inversion Theorem. Let b = Ax and c = 1
NAb. Claim that

c = x. By definition

bk =
N−1∑
l=0

e−i2πkl/Nxl and cp =
1

N

N−1∑
q=0

ei2πpq/Nbq.

Substituting yields

cp =

N−1∑
q=0

e−i2πpq/N
( 1

N

N−1∑
l=0

ei2πql/Nxl

)
=

1

N

N−1∑
l=0

{N−1∑
q=0

ei2π(l−p)q/N
}
xl

=
1

N

N−1∑
l=0

{
N for l = p

0 otherwise

}
xl =

N

N
xp = xp.

This finishes the proof of the theorem. ////

Let’s pause for a moment to implement a computer program that computes the Fourier
transform and the inverse Fourier transform directly from the definitions using matrix-
vector multiplication. The resulting Fortran code looks like

1 program main
2 implicit none
3 integer,parameter :: FTSIZE=8
4 real*8,parameter :: M_PI=3.141592653589793238460D0
5 integer :: i
6 complex*16,dimension(0:FTSIZE-1):: X,B,C
7 do i=0,FTSIZE-1
8 X(i)=cmplx(1D0/(i+1),1D0/(FTSIZE-i),kind(X(i)));
9 end do

10 print '("N=",I0)',FTSIZE
11 print '("X=")'
12 call cvecprint(X);
13 call dft(X,B)
14 print '("B=")'
15 call cvecprint(B)
16 call invdft(B,C)
17 print '("C=")'
18 call cvecprint(C)
19 return
20

21 contains
22

23 subroutine dft(x,b)
24 complex*16,dimension(0:) :: x,b
25 integer :: N,l,k
26 N=size(x)
27 do l=0,N-1
28 b(l)=0
29 end do
30 do k=0,N-1
31 do l=0,N-1
32 b(k)=b(k)+exp(cmplx(0D0,-2*M_PI*k*l/N,kind(b(k))))*x(l)
33 end do
34 end do
35 end subroutine
36

37 subroutine invdft(x,b)
38 complex*16,dimension(0:) :: x,b
39 integer :: N,l,k

40 N=size(x)
41 do l=0,N-1
42 b(l)=0
43 end do
44 do k=0,N-1
45 do l=0,N-1
46 b(k)=b(k)+exp(cmplx(0,2*M_PI*k*l/N,kind(b(k))))*x(l)/N;
47 end do
48 end do
49 end subroutine
50

51 subroutine cvecprint(x)
52 complex*16,dimension(0:) :: x
53 integer :: N,i
54 N=size(x)
55 do i=0,N-1
56 print '("(",G0," ",G0,")")',x(i)
57 end do
58 end subroutine
59

60 end program

and produces the output

N=8
X=
(1.0000000000000000 0.12500000000000000)
(0.50000000000000000 0.14285714285714285)
(0.33333333333333331 0.16666666666666666)
(0.25000000000000000 0.20000000000000001)
(0.20000000000000001 0.25000000000000000)
(0.16666666666666666 0.33333333333333331)
(0.14285714285714285 0.50000000000000000)
(0.12500000000000000 1.0000000000000000)
B=
(2.7178571428571425 2.7178571428571425)
(-0.86391851475668857E-001 -0.20856837951108187)
(0.22204460492503131E-015 -0.58333333333333370)
(0.28564698969660285 -0.68961283657658679)
(0.63452380952381005 -0.63452380952380949)
(1.0197251848090030 -0.42238400144129962)
(1.4476190476190478 -0.12767564783189300E-014)
(1.9810196769700616 0.82056521752897127)
C=
(1.0000000000000000 0.12500000000000014)

(0.49999999999999978 0.14285714285714307)
(0.33333333333333370 0.16666666666666718)
(0.25000000000000050 0.19999999999999968)
(0.20000000000000015 0.24999999999999944)
(0.16666666666666596 0.33333333333333359)
(0.14285714285714185 0.49999999999999989)
(0.12500000000000078 0.99999999999999989)

Note that the value for c is the same as x. This is consistent with the Fourier Inversion
Theorem and leads us to believe that the above code is producing correct results. Making
sure the code is producing the correct answer is an important first step before any sort of
optimization is attempted.

We now analyze the performance of the above simple Fourier transform code. Observe
that the dft and invdft routines each consist of two loops of length N . As the loops are
nested, the resulting number of operations is N2. We will obtain a significant performance
increase by changing the code to use the fast Fourier transform algorithm which only takes
about N log2 N number of operations. Before doing this, we instrument the above slow
algorithm with timing routines and also create a parallel version for an example of parallel
programming. The modified code looks like

1 program main
2 implicit none
3 integer,parameter :: FTSIZE=8192
4 integer :: tic_start,tic_finish,tic_rate
5 real*8 :: t
6 real*8,parameter :: M_PI=3.141592653589793238460D0
7 integer :: i
8 complex*16,dimension(0:FTSIZE-1):: X,B,C
9 do i=0,FTSIZE-1

10 X(i)=cmplx(1D0/(i+1),1D0/(FTSIZE-i),kind(X(i)));
11 end do
12 print '("N=",I0)',FTSIZE
13 call tic
14 call dft(X,B)
15 t=toc()
16 print '("B(0)=(",G0," ",G0,")")',B(0)
17 print '("dft took ",G0," seconds.")',t
18 !$omp parallel
19 !$omp single
20 call tic
21 call pdft(X,B)
22 t=toc()
23 !$omp end single
24 !$omp end parallel
25 print '("B(0)=(",G0," ",G0,")")',B(0)

26 print '("parallel dft took ",G0," seconds.")',t
27 return
28

29 contains
30

31 subroutine tic
32 call system_clock(count_rate=tic_rate)
33 call system_clock(tic_start)
34 end subroutine
35

36 real*8 function toc()
37 call system_clock(tic_finish)
38 toc=float(tic_finish-tic_start)/tic_rate
39 end function
40

41 subroutine dft(x,b)
42 complex*16,dimension(0:) :: x,b
43 integer :: N,l,k
44 N=size(x)
45 do l=0,N-1
46 b(l)=0
47 end do
48 do k=0,N-1
49 do l=0,N-1
50 b(k)=b(k)+exp(cmplx(0D0,-2*M_PI*k*l/N,kind(b(k))))*x(l)
51 end do
52 end do
53 end subroutine
54

55 subroutine pdft(x,b)
56 complex*16,dimension(0:) :: x,b
57 integer :: N,l,k
58 N=size(x)
59 do l=0,N-1
60 b(l)=0
61 end do
62 !$omp taskloop default(shared)
63 do k=0,N-1
64 do l=0,N-1
65 b(k)=b(k)+exp(cmplx(0D0,-2*M_PI*k*l/N,kind(b(k))))*x(l)
66 end do
67 end do
68 !$omp end taskloop
69 end subroutine

70

71 end program

To use multiple processor cores the only change needed is to add a parallel loop for the
matrix multiplication denoted by omp taskloop on line 62. On a 3.0Ghz dual-core AMD
A6-9225 based notebook computer the above code produces the output

N=8192
B(0)=(9.5881900460952654 9.5881900460953098)
dft took 4.6409997940063477 seconds.
B(0)=(9.5881900460952654 9.5881900460953098)
parallel dft took 2.4769999980926514 seconds.

For this system, a factor of 1.87 performance increase is obtained when switching from one
to two CPUs. The same program when run on a 2.4Ghz twelve-core Intel Xeon E5-2620
based system with hyperthreading enabled produces the output

N=8192
B(0)=(9.5881900460952654 9.5881900460953098)
dft took 5.9670000076293945 seconds.
B(0)=(9.5881900460952654 9.5881900460953098)
parallel dft took 0.49399998784065247 seconds.

In this case the performance increase was about 12 times faster which is essentially linear
with the number of cores. Sometimes performance increases sub-linearly as we will shall
see later.

The Fast Fourier Transform

While a factor 18 speedup was easy to obtain by parallelizing the slow algorithm, in the
case of the Fourier transform much more significant gains can be achieved by using a
conquer and divide approach. This is possible because the matrix A corresponding to the
Fourier transform has a significant number of symmetries in it based on the factors of the
length N of the transform. For simplicity we will assume that N = 2n for some positive
integer n. Thus, N is divisible by 2 and we can write 2K = N . It follows that

N−1∑
l=0

e−i2πkl/Nxl =
∑

l even
e−i2πkl/Nxl +

∑
l odd

e−i2πkl/Nxl

=

K−1∑
p=0

e−i2πkp/Kx2p + e−i2πk/N
K−1∑
p=0

e−i2πkp/Kx2p+1

(1)

Note that the original Fourier transform of size N has been rewritten as two smaller
Fourier transforms of size K which then need to be combined. The combining is done
by multiplying the second transform by the factor e−i2πk/N for k = 0, 1, . . . , N − 1 which
results in N additional multiplications. Therefore, the total number of operations has been
reduced to

K2 +N +K2 = 2
(N
2

)2

+N =
1

2
N2 +N

which is a reduction of almost half the original N2.
We are now ready to prove

The Fast Fourier Transform Theorem. Suppose N = 2n, then the Fourier transform
can be computed in N log2 N number of operations.

Proof of The Fast Fourier Transform Transform Theorem. Consider the minimal
number of operations Tn needed to perform a discrete Fourier transform of size 2n. By the
conquer and divide step described above, we know that

Tn ≤ 2Tn−1 + 2n and similarly Tn−1 ≤ 2Tn−2 + 2n−1.

Substituting the latter in to the former yields Tn ≤ 22Tn−2 + 2(2n) and by induction it
follows that

Tn ≤ 2nT0 + n2n.

Since the transform of length one is the identity then T0 = 0. Consequently, Tn ≤ N log2 N .
This shows the Fourier transform can be computed in N log2 N operations. ////

We remark that N log2 N number of operations can be much smaller than N2 when
N is large. When N = 8192, as used for our previous numerical test, it follows that

N log2 N = 106496 and N2 = 67108864.

Since 67108864/106496 ≈ 630, using the fast Fourier transform has the performance ad-
vantage of about 630 additional processor cores when N = 8192. For larger values of N
the advantages are even more pronounced. When N = 65536 the slow algorithm takes
an impractically long time; for values of N corresponding to vectors that are sized to the
limits of available memory, the fast algorithm is the only way to complete the computation.

We finish by presenting a recursive routine written in Fortran to compute the fast
Fourier transform. The code

1 program main
2 implicit none
3 integer,parameter :: FTSIZE=8
4 real*8,parameter :: M_PI=3.141592653589793238460D0
5 integer :: i
6 complex*16,dimension(0:FTSIZE-1):: X,B,C
7 do i=0,FTSIZE-1
8 X(i)=cmplx(1D0/(i+1),1D0/(FTSIZE-i),kind(X(i)));
9 end do

10 print '("N=",I0)',FTSIZE
11 call fft(X,B)
12 print '("fft_B=")'
13 call cvecprint(B)
14 return
15

16 contains
17

18 recursive subroutine fftwork(N,s,o,x,p,b)
19 integer :: k,N2,N,s,o,p
20 complex*16,dimension(0:) :: x,b
21 complex*16 :: even,odd,w
22 if(N.eq.1) then
23 b(p)=x(o)
24 return
25 end if
26 if(mod(N,2).ne.0) then
27 print '("N not divisible by 2!")'
28 stop
29 end if
30 N2=N/2
31 call fftwork(N2,2*s,o,x,p,b)
32 call fftwork(N2,2*s,o+s,x,p+N2,b)
33 do k=0,N2-1
34 even=b(p+k)
35 odd=b(p+k+N2)
36 w=exp(cmplx(0.0D0,-2*M_PI*k/N,kind(w)))
37 b(p+k)=even+w*odd
38 b(p+k+N2)=even-w*odd
39 end do
40 end subroutine
41

42 subroutine fft(x,b)
43 complex*16,dimension(0:) :: x,b
44 integer :: N
45 N=size(x)
46 call fftwork(N,1,0,x,0,b)
47 end subroutine
48

49 subroutine cvecprint(x)
50 complex*16,dimension(0:) :: x
51 integer :: N,i
52 N=size(x)
53 do i=0,N-1
54 print '("(",G0," ",G0,")")',x(i)
55 end do
56 end subroutine
57

58 end program

produces the output

N=8
fft_B=
(2.7178571428571425 2.7178571428571425)
(-0.86391851475668746E-001 -0.20856837951108154)
(0.0000000000000000 -0.58333333333333326)
(0.28564698969660318 -0.68961283657658712)
(0.63452380952380949 -0.63452380952380949)
(1.0197251848090021 -0.42238400144129939)
(1.4476190476190476 0.55511151231257827E-016)
(1.9810196769700634 0.82056521752896805)

Compare the output for the slow routine to the fast routine. When optimizing an computer
program it is important to compared results produced by the optimized code to known
correct results. The fact that the output is the same in this case, suggests that the
optimized code performs the same calculation as the original program. Although one test
case—or even a hundred—would not be sufficient guarantee two different algorithms always
produce the same results, such testing is useful and can catch many errors.

For now, we assume the code is correct and proceed to check performance by instru-
menting the code with timing routines and also creating a parallel version as was done for
the slow Fourier transform. The modified code looks like

1 program main
2 implicit none
3 integer,parameter :: FTSIZE=1048576
4 integer :: tic_start,tic_finish,tic_rate
5 real*8 :: t
6 real*8,parameter :: M_PI=3.141592653589793238460D0
7 integer :: i
8 complex*16,dimension(:),allocatable:: X,B
9 allocate(X(0:FTSIZE-1))

10 allocate(B(0:FTSIZE-1))
11 do i=0,FTSIZE-1
12 X(i)=cmplx(1D0/(i+1),1D0/(FTSIZE-i),kind(X(i)));
13 end do
14 print '("N=",I0)',FTSIZE
15 call tic
16 call fft(X,B)
17 t=toc()
18 print '("B(0)=(",G0," ",G0,")")',B(0)
19 print '("fft took ",G0," seconds.")',t
20 !$omp parallel
21 !$omp single
22 call tic
23 call pfft(X,B)
24 t=toc()

25 !$omp end single
26 !$omp end parallel
27 print '("B(0)=(",G0," ",G0,")")',B(0)
28 print '("parallel fft took ",G0," seconds.")',t
29 return
30

31 contains
32

33 subroutine tic
34 call system_clock(count_rate=tic_rate)
35 call system_clock(tic_start)
36 end subroutine
37

38 real*8 function toc()
39 call system_clock(tic_finish)
40 toc=float(tic_finish-tic_start)/tic_rate
41 end function
42

43 recursive subroutine fftwork(N,s,o,x,p,b)
44 integer :: k,N2,N,s,o,p
45 complex*16,dimension(0:) :: x,b
46 complex*16 :: even,odd,w
47 if(N.eq.1) then
48 b(p)=x(o)
49 return
50 end if
51 if(mod(N,2).ne.0) then
52 print '("N not divisible by 2!")'
53 stop
54 end if
55 N2=N/2
56 call fftwork(N2,2*s,o,x,p,b)
57 call fftwork(N2,2*s,o+s,x,p+N2,b)
58 do k=0,N2-1
59 even=b(p+k)
60 odd=b(p+k+N2)
61 w=exp(cmplx(0D0,-2*M_PI*k/N,kind(w)))
62 b(p+k)=even+w*odd
63 b(p+k+N2)=even-w*odd
64 end do
65 end subroutine
66

67 subroutine fft(x,b)
68 complex*16,dimension(0:) :: x,b

69 integer :: N
70 N=size(x)
71 call fftwork(N,1,0,x,0,b)
72 end subroutine
73

74 recursive subroutine pfftwork(N,s,o,x,p,b)
75 integer :: k,N2,N,s,o,p
76 complex*16,dimension(0:) :: x,b
77 complex*16 :: even,odd,w
78 if(N.eq.1) then
79 b(p)=x(o)
80 return
81 end if
82 if(mod(N,2).ne.0) then
83 print '("N not divisible by 2!")'
84 stop
85 end if
86 N2=N/2
87 !$omp task default(shared) final(s<=32)
88 call pfftwork(N2,2*s,o,x,p,b)
89 !$omp end task
90 call pfftwork(N2,2*s,o+s,x,p+N2,b)
91 !$omp taskwait
92 do k=0,N2-1
93 even=b(p+k)
94 odd=b(p+k+N2)
95 w=exp(cmplx(0D0,-2*M_PI*k/N,kind(w)))
96 b(p+k)=even+w*odd
97 b(p+k+N2)=even-w*odd
98 end do
99 end subroutine

100

101 subroutine pfft(x,b)
102 complex*16,dimension(0:) :: x,b
103 integer :: N
104 N=size(x)
105 call pfftwork(N,1,0,x,0,b)
106 end subroutine
107

108 subroutine cvecprint(x)
109 complex*16,dimension(0:) :: x
110 integer :: N,i
111 N=size(x)
112 do i=0,N-1

12



113 print '("(",G0," ",G0,")")',x(i)
114 end do
115 end subroutine
116

117 end program

For the parallel code omp task directive in line 87 schedules one of the recursive calls to
compute a smaller Fourier transform in a separate worker thread while the current thread
recursively computes the other Fourier transform. The omp taskwait on line 91 makes sure
both of the recursive calls have completed before the results are combined with a parallel
loop in line 46. After pfft recurses 5 times, the stride given by s is equal 32 and 32 parallel
tasks have been created to perform the computation. At this point the final clause causes
the rest of the recursion to be made in serial. On a 3.0Ghz dual-core AMD A6-9225 based
notebook computer the above code produces the output

N=1048576
B(0)=(14.440159752937522 14.440159752937522)
fft took 0.72000002861022949 seconds.
B(0)=(14.440159752937522 14.440159752937522)
parallel fft took 0.48100000619888306 seconds.

We note that the fast Fourier transform performed a transform of size N = 1048576 faster
than the slow algorithm could handle a transform of size N = 8192. This time only a factor
of 1.5 parallel speedup occurs when working computing two cores. The same program when
run on a 2.4Ghz twelve-core Intel Xeon E5-2620 based system produces the output

N=1048576
B(0)=(14.440159752937522 14.440159752937522)
fft took 0.82099997997283936 seconds.
B(0)=(14.440159752937522 14.440159752937522)
parallel fft took 0.20800000429153442 seconds.

In this case the performance increase was about 4 times faster, which is only 33 percent of
the optimal 12 times speedup. Parallel efficiency can be limited by the read-write band-
width of main memory as well as overhead related to scheduling the parallel loop on the
multiple cores. We observe that the memory access patterns of the fast Fourier transform
involve recursively skipping by odd and even indexes. In particular the fast Fourier trans-
form does not access memory sequentially. It is likely that this creates additional pressure
on the memory subsystem compared to the simple discrete transform discussed earlier.

Homework Problems

1. The conquer and divide step described in equation (1) splits the terms of the sum for
the discrete Fourier transform into odd and even terms. Construct a similar equation
for use when N = 3n that divides the sum into three parts such that l divided by 3
has remainder 0, 1 or 2.

2. Let z = a + bi and w = u + iv be complex numbers. It takes four real-valued
multiplications when using the foil method to find the product zw. Look up fast
complex multiplication, describe it and explain how many real-valued multiplications
the fast algorithm uses to find zw.

3. Compute the number of real-valued double-precision floating point multiplications and
additions per second achieved for test runs of the the fast Fourier transform detailed
above. Explain your reasoning and how you counted the total number of operations.
How many evaluations of the exponential function are performed?

4. Download the code fasttime.f90 for determining the speed of the fast Fourier trans-
form from our website. Compile and run it on your computer. Compare the speed of
this code to the one developed in class.

5. [Extra Credit] Constuct an improved Fourier transform code that runs faster than
either the ones presented in this handout or the one developed in class. Consider
evaluating the exponential function ahead of time. Another optimization would be to
avoid letting the routine recurse all the way down to the identity transform of length
one and instead end with an optimized transform of length 4 or 8.

